

# Wetzel: Formalisation of an Undecidable Problem Linked to the Continuum Hypothesis

Lawrence C Paulson<sup>[0000–0003–0288–4279]</sup>

Computer Laboratory, University of Cambridge, UK  
lp15@cam.ac.uk <https://www.cl.cam.ac.uk/~lp15/>

**Abstract.** In 1964, Paul Erdős published a paper [5] settling a question about function spaces that he had seen in a problem book. Erdős proved that the answer was yes if and only if the continuum hypothesis was false: an innocent-looking question turned out to be undecidable in the axioms of ZFC. The formalisation of these proofs in Isabelle/HOL demonstrate the combined use of complex analysis and set theory, and in particular how the Isabelle/HOL library for ZFC [17] integrates set theory with higher-order logic.

**Keywords:** Isabelle· Erdős· continuum hypothesis· set theory· complex analysis· formalisation of mathematics

## 1 Introduction

This story [6] expresses the richness of mathematics. It seems that Paul Erdős found the following question in a problem book belonging to the mathematics department of Ann Arbor University:

Suppose that  $F$  is a family of analytic functions on  $\mathbb{C}$  such that for each  $z$  the set  $\{f(z) : f \in F\}$  is countable. (Call this *property*  $P_0$ .) Then is the family  $F$  itself countable?

This question apparently arose in the PhD work of John E. Wetzel, in connection with spaces of harmonic functions. Erdős was able to show that if the continuum hypothesis failed, every family satisfying  $P_0$  had to be countable; but if the hypothesis held, there existed (by a transfinite construction) an uncountable family satisfying  $P_0$ . His proof appears in *Proofs from THE BOOK* [2],<sup>1</sup> Aigner and Ziegler’s collection of “perfect proofs” inspired by Erdős.

Cantor’s celebrated *continuum hypothesis* (CH), number one on Hilbert’s list of fundamental questions, asks whether there exists a cardinality between that of the natural numbers, namely  $\aleph_0$ , and that of the real numbers, namely  $\mathfrak{c}$ . Since the next cardinal after  $\aleph_0$  is denoted  $\aleph_1$  and the cardinality of the continuum is known to equal  $2^{\aleph_0}$ , CH can be written symbolically as  $\aleph_1 = 2^{\aleph_0}$ . It’s fundamental because the notion of a countable set is straightforward, as is the proof by diagonalisation that the real numbers cannot be enumerated (Cantor’s

<sup>1</sup> End of Chapter 19. Also [https://doi.org/10.1007/978-3-662-57265-8\\_19](https://doi.org/10.1007/978-3-662-57265-8_19)

theorem). CH asserts that no set exists of intermediate size between the natural numbers and the reals. It was shown to be consistent with the axioms of set theory by Gödel and to be independent from them by Cohen. The details can be found in any set theory text [11].

These results have been machine verified by a variety of authors. Han and van Doorn [10] were the first to mechanise Cohen’s proof of the independence of CH, using Lean.<sup>2</sup> I have formalised Gödel’s model, the *constructible sets* [16], and some of its properties. Gunther et al. [9] have formalised model constructions both to confirm and refute CH using forcing. These latter formalisations were done using Isabelle/ZF [20], the instance of Isabelle for first-order logic and set theory. However, Isabelle/HOL is much better developed than Isabelle/ZF; in particular, the Wetzel problem requires its complex analysis library. So this paper demonstrates how to tackle a problem that combines the worlds of analysis and set theory, including such mysteries as holomorphic functions, transfinite cardinals and recursion up to uncountable ordinals, on the common basis of higher-order logic.

The paper outlines the development of formal treatments of ZF set theory in Isabelle/HOL (§2), continuing to the Wetzel problem in the non-CH case (§3) and the more difficult CH case (§4). Next comes some discussion of the work in the context of the formalisation of mathematics generally (§5), and finally conclusions.

The formal development itself is available online [19] in Isabelle’s Archive of Formal Proofs (AFP), as is the ZFC formalisation [17] on which it is built.

## 2 Isabelle and Set Theory

Isabelle is a generic theorem prover, ultimately based on a uniform representation of logical syntax in the typed lambda calculus, with inference rules expressed syntactically and combined using higher-order unification [14]. Although Isabelle/HOL [12]—the version for higher-order logic—is by far the best known and most developed instance of Isabelle, other instances exist. These include Isabelle/FOL (classical first-order logic) and Isabelle/ZF. The latter is a faithful development of set theory from the Zermelo–Fraenkel axioms within first-order logic. Since the axiom of choice (AC) is kept separate from the other axioms, one can investigate weaker forms of AC and equivalents of AC [20].

During the 1990s, I made some significant formal developments using Isabelle/ZF [15], culminating in a proof of the relative consistency of the axiom of choice using Gödel’s constructible universe [16]. Recently, several highly impressive formalisations of forcing were done within Isabelle/ZF [8,9].

On the other hand, Isabelle/ZF lacks much of the automation found in Isabelle/HOL and has no theory even of the real numbers. That makes it unsuitable for the Wetzel problem. However, it is also possible to formalise ZF set theory within higher-order logic.

<sup>2</sup> <https://leanprover.github.io>

## 2.1 Formalising ZF in Higher-Order Logic

During the 1990s, Michael JC Gordon conducted several experiments [7] involving the formalisation of ZF set theory in his HOL proof assistant. His HOL-ST simply introduced a type  $\mathbf{V}$  of all sets and a relation  $\in : \mathbf{V} \times \mathbf{V} \rightarrow \text{bool}$ , then asserted all the Zermelo-Fraenkel axioms. Sten Agerholm [1] formalised Dana Scott’s inverse limit construction of the set  $D_\infty$ , satisfying  $D_\infty \cong [D_\infty \rightarrow D_\infty]$  and yielding a model of the untyped  $\lambda$ -calculus.

The use of higher-order logic as opposed to first-order logic as the basis makes this version of set theory somewhat stronger than standard ZF. In HOL-ST it is possible to define the syntax of first-order logic and the semantics of set-theoretic formulas in terms of  $\mathbf{V}$ , verifying the ZF axioms and thus proving their consistency. On the other hand, a model for HOL-ST can be constructed in ZF plus one inaccessible cardinal. These remarks (which Gordon credited to Kenneth Kunen) together imply that the strength of HOL-ST is somewhere between ZF and ZF plus one inaccessible cardinal. This is a weak assumption, much weaker than the dependent type theories used in Coq and Lean, which are stronger than any finite number of inaccessible cardinals [23].

Some time later, Steven Obua performed a similar experiment [13], using Isabelle/HOL. He adopted the same axioms and overall approach as Gordon, and demonstrated his system by formalising John H Conway’s *partizan games* [21]. Obua obtained some interoperability with the existing infrastructure of Isabelle/HOL, such as its recursive function definitions.

## 2.2 Axiomatic Type Classes

Deeper integration requires the use of axiomatic type classes, introduced by Wenzel in 1997. An *axiomatic type class* [22] defines an open-ended collection of types on the basis of a signature and a possibly empty list of axioms. The signature specifies certain operations and their types, which will be polymorphic with type variables referring to that type class. The operations may also be equipped with concrete syntax, such as infix declarations. Any type, whether already existing or defined in the future, can be shown to be an instance of the class if it provides definitions of all the operations in the signature that satisfy the associated axioms.

Isabelle/HOL defines a series of type classes for orderings:

- *ord* introduces the relations  $\leq$  and  $<$ , of type  $'a \Rightarrow 'a \Rightarrow \text{bool}$  but no axioms
- *preorder* extends *ord* with axioms for reflexivity and transitivity
- *order* extends *preorder* with antisymmetry
- *linorder* extends *order* with the axiom  $x \leq y \vee y \leq x$ , for linear orderings

The arithmetic types *nat*, *int*, *real* belong to each of these classes; the type operator  $\times$  preserves membership in each of these classes.

### 2.3 The development ZFC-in-HOL

My own framework [17] for set theory within Isabelle/HOL is mathematically equivalent to Gordon's and Obua's. Its aims were practical:

1. To reproduce as much of Isabelle/ZF as possible,
2. while achieving maximum integration with the underlying higher-order logic.

To achieve these aims I relied on axiomatic type classes, whenever possible overloading existing symbols for set theory rather than introducing new vocabulary.

It introduces the type  $V$  of sets and the function  $elts$  of type  $V \Rightarrow V\ set$  mapping a set to its elements. Thus it uses Isabelle/HOL's existing typed sets to represent classes, but not all elements of  $V\ set$  correspond to sets. The predicate *small* identifies those that are small enough, but here comes the first stage of integration: *small* needs to be polymorphic, accepting a set of any type and with the more general meaning that a set is small if its elements can be put into one-to-one correspondence with the elements of a ZF set. A set that is small in this more general sense does not in itself denote a ZF set, but this condition can be necessary if it is used in constructions that ultimately lead to a ZF set.

The fundamentals of set theory are built up as usual, starting with the ZF axioms. Set membership is expressed using the existing (typed) set membership operator:  $x \in elts\ y$ . Union, intersection and the subset relation are expressed using the type class *distrib\_lattice* (distributive lattices), which already provides the symbols  $\sqcup$ ,  $\sqcap$ ,  $\leq$ , etc. Unions and intersections of families rely on the type class *conditionally\_complete\_lattice*, which provides the symbols  $\bigsqcup$  and  $\bigsqcap$ . Type classes also allow the overloading of  $0$  to denote the empty set (which is also the natural number zero) and  $1$  for the natural number one.

On this foundation, it was easy to import significant chunks of Isabelle/ZF: above all, cardinal arithmetic and the  $\aleph$  operator. It was also possible to reuse Isabelle/HOL's existing theory of recursion to obtain transfinite recursion on ordinals, and hence order types, Cantor normal form and much else. So the first aim was met, but more needed to be done to achieve the second.

### 2.4 The integration of ZFC-in-HOL with Isabelle/HOL

In theory, ZFC suffices for the formalisation of all the mathematics in this problem, and much more. In practice, we absolutely do not want to be forced to develop complex analysis from first principles in set theory when it already exists in the Isabelle/HOL libraries. And while transfinite cardinalities and other constructions are typically understood from the framework of ZFC, they are perfectly intelligible in a broader context.

The simplest generalisation is for cardinality. The *cardinality*  $|x|$  of a ZF set  $x$  is simply the minimal ordinal that is equipollent to<sup>3</sup>  $x$ . This definition generalises naturally to sets of any type. We can now refer to the cardinality of sets of real and complex numbers.

---

<sup>3</sup> in one-to-one correspondence with

Next comes *transfinite recursion*, also known as  $\epsilon$ -recursion. It is a sort of fixedpoint operator allowing the definition of functions over the whole of  $\mathbf{V}$ . If a function  $H$  is given, then transfinite recursion yields  $F$  such that for any  $a$ ,

$$F(a) = H(F \upharpoonright a), \quad (1)$$

where  $F \upharpoonright a$  denotes  $F$  itself, restricted to the elements of  $a$ . It is just an instance of well-founded recursion on the membership relation. The definition is simple:

**definition** *transrec* :: " $(V \Rightarrow 'a) \Rightarrow V \Rightarrow 'a \Rightarrow V \Rightarrow 'a$ "  
**where** "*transrec*  $H a \equiv wfrec \{(x,y). x \in elts y\} H a$ "

This version differs from the original [17] only in its type, which is now polymorphic as shown, allowing the recursively defined function to return anything. Here, *wfrec* is Isabelle/HOL's built-in operator for well-founded recursion. The recursion equation (1) easily follows.

**lemma** *transrec*: "*transrec*  $H a = H (\lambda x \in elts a. transrec H x) a$ "

Although transfinite recursion is typically used to define operations within the set-theoretic universe  $\mathbf{V}$ , we can now use it to create an uncountable set of analytic functions.

## 2.5 Embedding Isabelle/HOL Types into $\mathbf{V}$

The ZFC-in-HOL library defined the class of types that can be *embedded* into the set theoretic universe,  $V$ , by some injective map,  $V\_of$ :

**class** *embeddable* =  
**assumes** *ex\_inj*: " $\exists V\_of :: 'a \Rightarrow V. inj V\_of$ "

If a type is embeddable then each of its elements corresponds to some ZF set.

As it happens, Isabelle/HOL already provides the class *countable* of all types that can be embedded into the natural numbers. The latter are trivially embedded into  $V$  as finite ordinals, so it is easy to show that all countable types are *embeddable*. Examples include *nat*, *int*, *rat*, *bool*. Trivially,  $V$  can be embedded into itself, and the type constructors  $\times$ ,  $+$  and *list* are straightforwardly shown to preserve the embeddable property.

The library also defines the class of types that are small, which means that the type itself corresponds to some ZF set. It is defined in terms of the predicate *small*:

**class** *small* =  
**assumes** *small*: "*small* (*UNIV*::' $a$  set)"

Every *countable* type (and in particular those listed above) is *small*, because  $\omega$  is a set. And it's obvious that every *small* type is *embeddable*. By proving types to be small, we further extend the embeddable class. The type constructors  $\times$ ,  $+$  and *list* preserve smallness. The situation for the function type constructor ( $\rightarrow$ ) is a little subtle:

```

instance "fun" :: (small,embeddable) embeddable
instance "fun" :: (small,small) small

```

The (straightforward) proofs are omitted. It should be obvious that we cannot expect to embed  $\mathbf{V} \rightarrow \mathbf{V}$  into  $\mathbf{V}$ , but any function  $f \in A \rightarrow \mathbf{V}$  must be a set provided  $A$  is a set.

Types *real* and *complex* are also *small*. How do we know? The reals are obtained by quotienting type *nat*  $\Rightarrow$  *rat*. The details of their construction do not concern us. This is the full text of the proof.

```

instance real :: small
proof -
  have "small (range (Rep_real))"
    by simp
  then show "OFCLASS(real, small_class)"
    by intro_classes
      (metis Rep_real_inverse image_inv_f_f inj_on_def replacement)
qed

```

And type *complex* is essentially the same thing as  $\mathbb{R}^2$ .

The type classes *embeddable* and *small* aren't especially visible in the formalisation below. However, they provide the crucial link between type *V* and other Isabelle/HOL types. They are the key to combining set theory and complex analysis. For example, the check that some entity is small (representable by a ZF set) is usually automatic, thanks to this type class setup. Because *real* and *complex* are small, we can reason about their cardinality.

### 3 Wetzel's Problem: the $\neg$ CH Case

Now it is time to formalise the proof itself, following Aigner and Ziegler's presentation [2]. We begin by defining the predicate *Wetzel*, corresponding to  $P_0$  above, on sets of complex-valued functions. It holds if every element of the given set  $F$  is analytic on the complex plane and if, for all  $z$ , the set  $\{f(z) : f \in F\}$  is countable:

```

definition Wetzel :: "(complex  $\Rightarrow$  complex) set  $\Rightarrow$  bool"
  where "Wetzel  $\equiv$   $\lambda F. (\forall f \in F. f \text{ analytic\_on UNIV}) \wedge (\forall z. \text{countable}((\lambda f. f z) ` F))"$ 

```

Remarkably, the Isabelle/HOL proof is barely 50 lines. First, here is the theorem statement:

```

proposition Erdos_Wetzel_nonCH:
assumes W: "Wetzel F" and NCH: "C_continuum  $>$   $\aleph_1$ "
shows "countable F"

```

We set out to prove the contrapositive, negating the conclusion:

```

have " $\exists z0. \text{gcard} ((\lambda f. f z0) ` F) \geq \aleph_1$ " if "uncountable F"

```

Given the uncountable family  $F$ , we find a subset  $F' \subseteq F$  of cardinality  $\aleph_1$  and thus a bijection  $\phi : \aleph_1 \rightarrow F'$  between the ordinals below  $\aleph_1$  and  $F'$ .

```

have "gcard F ≥ ℵ1"
  using that uncountable_gcard_ge by force
then obtain F' where "F' ⊆ F" and F': "gcard F' = ℵ1"
  by (meson Card_Aleph_subset_smaller_gcard)
then obtain φ where φ: "bij_betw φ (elts ω1) F'"
  by (metis TC_small_eqpoll_def_gcard_eqpoll)

```

We next define the family of sets  $S(\alpha, \beta)$  as  $\{z. \phi_\alpha(z) = \phi_\beta(z)\}$ . Here  $\alpha$  and  $\beta$  range over ordinals, and  $\omega_1$  is the first uncountable ordinal (thus the same set as  $\aleph_1$ , but regarded as an ordinal). The  $\phi_\alpha$  for  $\alpha < \omega_1$  are the given analytic functions.

```

define S where "S ≡ λα β. {z. φ α z = φ β z}"

```

It takes 10 lines to prove that  $S(\alpha, \beta)$  is countable for  $\alpha < \beta < \omega_1$ , since two distinct holomorphic functions on  $\mathbb{C}$  can agree at only countably many arguments (a fact proved by Aigner and Ziegler).

```

have "gcard (S α β) ≤ ℵ0" if "α ∈ elts β" "β ∈ elts ω1" for α β

```

The next step is to define  $SS$  as the union of all  $S(\alpha, \beta)$  for  $\alpha < \beta < \omega_1$ .

```

define SS where "SS ≡ ⋃β ∈ elts ω1. ⋃α ∈ elts β. S α β"

```

A 14 line calculation shows that  $|SS| \leq \aleph_1$ , but we are assuming the negation of CH, so  $SS$  cannot be the entire complex plane: there exists some  $z_0 \notin SS$ .

```

finally have "gcard SS ≤ ℵ1" .
with NCH obtain z0 where "z0 ∉ SS"
  by (metis Complex_gcard_UNIV_eq_I less_le_not_le)

```

That  $z_0$  satisfies our requirements follows straightforwardly by the definitions of  $S$  and  $SS$ .

## 4 Wetzel's Problem: the CH Case

Assuming CH, it is possible to construct an uncountable family of analytic functions that makes the Wetzel property  $P_0$  fail. It's a rather delicate transfinite recursion, so let's review the argument before examining the formal proof.

### 4.1 The transfinite construction

CH implies that  $|\mathbb{C}| = \aleph_1$  and we can write  $\mathbb{C} = \{\zeta_\alpha : \alpha < \omega_1\}$ . Now consider the set  $D \subseteq \mathbb{C}$  of *rational* complex numbers:

$$D = \{p + iq : p, q \in \mathbb{Q}\}.$$

Suppose that we had a family of functions  $\{f_\beta : \beta < \omega_1\}$  such that

$$f_\beta(\zeta_\alpha) \in D \quad \text{if } \alpha < \beta. \quad (2)$$

Since the set  $D$  is countable and the  $\zeta_\alpha$  for  $\alpha < \omega_1$  include all the complex numbers, the desired result would follow.

Erdős showed how to construct this family by transfinite induction. For an arbitrary  $\gamma < \omega_1$ , assume that a family of distinct analytic functions  $\{f_\beta : \beta < \gamma\}$  is defined below  $\gamma$ . To conclude the inductive argument and therefore prove the theorem, we must extend it with a new function,  $f_\gamma$ .

Since  $\gamma$  is countable, the set  $\{f_\beta : \beta < \gamma\}$  can be enumerated as  $\{g_0, g_1, \dots\}$  and  $\{\zeta_\alpha : \alpha < \gamma\}$  can be enumerated as  $\{w_0, w_1, \dots\}$ ; both sets are finite or infinite according to whether  $\gamma$  itself is finite or infinite. The sought-for analytic function  $f_\gamma$  should satisfy, for all  $n$ ,

$$f_\gamma(w_n) \in D \quad \text{and} \quad f_\gamma(w_n) \neq g_n(w_n). \quad (3)$$

The second condition above ensures that  $f_\gamma$  is new by diagonalisation, while the first is simply (2). We construct  $f_\gamma$  by putting

$$\begin{aligned} f_\gamma(z) := & \epsilon_0 + \epsilon_1(z - w_0) + \epsilon_2(z - w_0)(z - w_1) \\ & + \epsilon_3(z - w_0)(z - w_1)(z - w_2) + \dots \end{aligned} \quad (4)$$

Again, this sum is finite or infinite according to  $\gamma$ . The  $\{\epsilon_m\}$  are complex numbers chosen one at a time to satisfy conditions (3) above; ensuring that  $f_\gamma(w_n)$  avoids  $g_n(w_n)$  is possible because  $D$  is dense in  $\mathbb{C}$ . In the finite case the sum is a polynomial, so trivially analytic. In the infinite case, the choice of the  $\{\epsilon_m\}$  needs to be carefully calibrated in order to satisfy the conditions while converging to zero sufficiently rapidly. We have a lot of leeway and I chose

$$|\epsilon_m| < [m! \cdot \prod_{i < m} (1 + |w_i|)]^{-1}. \quad (5)$$

The summation (4) converges to an analytic function because the uniform limit of holomorphic functions is holomorphic, and the limit can be shown to be uniform by the Weierstrass  $M$ -test.<sup>4</sup>

A clever aspect of the construction is that the conditions (3) constrain only  $f_\gamma(w_n)$ , whose value depends only on  $\epsilon_m$  for  $m \leq n$ . That is,  $f_\gamma(w_0) = \epsilon_0$ ,  $f_\gamma(w_1) = \epsilon_0 + \epsilon_1(w_1 - w_0)$ , etc. The desired values of  $\epsilon_m$  can be calculated sequentially.

## 4.2 The Isabelle/HOL Formalisation

The full proof is some 280 lines of Isar text. Let's examine some highlights. First, let's state the theorem formally:

<sup>4</sup> Thanks to Manuel Eberl for suggesting this argument.



**proposition** *Erdos\_Wetzel\_CH*:  
**assumes** *CH*: " $C_{\text{continuum}} = \aleph_1$ "  
**obtains** *F* **where** "*Wetzel F*" **and** "*uncountable F*"

The proof begins with a self-evident definition of  $D$ . It's then shown to be countably infinite.

```
define D where "D  $\equiv \{z. \text{Re } z \in \mathbb{Q} \wedge \text{Im } z \in \mathbb{Q}\}$ "
have Def: "D =  $(\bigcup x \in \mathbb{Q}. \bigcup y \in \mathbb{Q}. \{\text{Complex } x \ y\})$ "
using complex.collapse by (force simp: D_def)
with countable_rat have "countable D"
by blast
have "infinite D"
  <proof>
have " $\exists w. \text{Re } w \in \mathbb{Q} \wedge \text{Im } w \in \mathbb{Q} \wedge \text{norm } (w - z) < e$ " if " $e > 0$ " for  $z \ e$ 
  <proof>
then have cloD: "closure D = UNIV"
by (auto simp: D_def closure_approachable dist_complex_def)
```

The closure of  $D$  equals the universal set  $UNIV$  of type *complex set*. We obtain the transfinite enumeration  $\{\zeta_\alpha : \alpha < \omega_1\}$  of the complex plane.

```
obtain  $\zeta$  where  $\zeta$ : "bij_betw  $\zeta$  (elts  $\omega_1$ ) (UNIV::complex set)"
```

Next come some technical definitions:  $inD$  for functions whose range for certain arguments lies within  $D$  and  $\Phi$  to express that  $f$  is a family of analytic functions indexed by the ordinals up to  $\beta$ .

```
define inD where "inD  $\equiv \lambda \beta f. (\forall \alpha \in \text{elts } \beta. f(\zeta \ \alpha) \in D)$ "
define " $\Phi \equiv \lambda \beta f. f \ \beta$  analytic_on UNIV  $\wedge$  inD  $\beta$  ( $f \ \beta$ )  $\wedge$  inj_on  $f$  (elts(succ  $\beta$ ))"
```

### 4.3 The Transfinite Construction

The following lemma is the step of the transfinite induction. A family  $f$  defined below the ordinal  $\gamma$  is extended with a new function,  $f_\gamma$  (the variable  $h$  below).

```
have *: " $\exists h. \Phi \ \gamma$  (restrict  $f$  (elts  $\gamma$ ))( $\gamma := h$ )"
```

```
if  $\gamma$ : " $\gamma \in \text{elts } \omega_1$ " and " $\forall \beta \in \text{elts } \gamma. \Phi \ \beta \ f$ " for  $\gamma \ f$ 
```

The construction of  $h$  depends on whether or not  $\gamma$  is finite:

```
obtain  $h$  where "h analytic_on UNIV" "inD  $\gamma \ h$ " " $\forall \beta \in \text{elts } \gamma. h \neq f \ \beta$ "
proof (cases "finite (elts  $\gamma$ )")
```

The finite case is easier, since the function we have to construct will simply be a polynomial, and trivially analytic. The finite ordinal  $\gamma$  is simply some natural number  $n$ , and the construction of  $f_\gamma$  is by induction on  $n$ .

Let's examine the infinite case, in which similar ideas are taken to the max. We formalise the step of writing  $\{f_\beta : \beta < \gamma\}$  as  $\{g_0, g_1, \dots\}$  and  $\{\zeta_\alpha : \alpha < \gamma\}$  as  $\{w_0, w_1, \dots\}$  by picking some bijection  $\eta$  between  $\mathbb{N}$  and  $\gamma$ , a countable ordinal which of course equals  $\{\alpha : \alpha < \gamma\}$ .

```

case False
then obtain  $\eta$  where  $\eta$ : "bij_betw  $\eta$  (UNIV::nat set) (elts  $\gamma$ )"
  by (meson  $\gamma$  countable_infiniteE' less_omega1_imp_countable)
define  $g$  where " $g \equiv f \circ \eta$ "
define  $w$  where " $w \equiv \zeta \circ \eta$ "

```

The following three definitions set up the finite approximants of our new analytic function,  $h$ . Note that  $p(n)$  is the product of  $z - w_i$  for  $i < n$ , while  $q(n)$  is  $\prod_{i < n} (1 + |w_i|)$ .

```

define  $p$  where " $p \equiv \lambda n z. \prod_{i < n}. z - w\ i$ "
define  $q$  where " $q \equiv \lambda n. \prod_{i < n}. 1 + \text{norm } (w\ i)$ "
define  $h$  where " $h \equiv \lambda n \epsilon z. \sum_{i < n}. \epsilon\ i * p\ i\ z$ "

```

The following three definitions constrain the next value  $\epsilon_n$ , a calculation left unspecified by previous authors [2,5]. To ensure convergence, each approximant needs to be close to the previous one, hence a neighbourhood (open ball) whose radius has  $n!q(n)$  for its denominator.  $DD$  is the intersection of this neighbourhood with the set  $D$ , excluding the one point that must be avoided. Finally, the function  $dd$  picks an arbitrary element of  $DD$ . In each of these functions, the argument  $\epsilon$  is an integer-valued function representing the  $\{\epsilon_i\}$  for  $i < n$ .

```

define BALL where
  "BALL  $\equiv \lambda n \epsilon. \text{ball } (h\ n\ \epsilon\ (w\ n))\ (\text{norm } (p\ n\ (w\ n)) / (\text{fact } n * q\ n))$ "
define DD where "DD  $\equiv \lambda n \epsilon. D \cap \text{BALL } n\ \epsilon - \{g\ n\ (w\ n)\}$ "
define dd where "dd  $\equiv \lambda n \epsilon. \text{SOME } x. x \in DD\ n\ \epsilon$ "

```

Because  $D$  is dense in  $\mathbb{C}$ , the set  $DD$  is always nonempty. Therefore,  $dd$  always chooses an element satisfying the constraints described above.

```

have "DD  $n\ \epsilon \neq \{\}$ " for  $n\ \epsilon$ 
  (proof)
then have dd_in_DD: "dd  $n\ \epsilon \in DD\ n\ \epsilon$ " for  $n\ \epsilon$ 
  by (simp add: dd_def some_in_eq)

```

Each  $\epsilon_n$  (written *coeff*  $n$  in Isabelle) is defined in terms of the previous epsilons by course of values recursion. The ugly details of the definition are omitted but the following line shows the recursion that it satisfies; note that  $h\ n\ \text{coeff } (w\ n)$  refers recursively to all the  $\epsilon_i$  for  $i < n$ . The last line verifies the bound (5).

```

define coeff where "coeff  $\equiv \dots$ "
have coeff_eq: "coeff  $n = (dd\ n\ \text{coeff} - h\ n\ \text{coeff } (w\ n)) / p\ n\ (w\ n)$ " for  $n$ 
  by (simp add: def_wfrec [OF coeff_def])
have norm_coeff: "norm (coeff  $n$ )  $< 1 / (\text{fact } n * q\ n)$ " for  $n$ 
  (proof)

```

To conclude, define the function  $f_\gamma$  (calling it *hh*) and prove that it is analytic.

```

define hh where "hh  $\equiv \lambda z. \text{suminf } (\lambda i. \text{coeff } i * p \ i \ z)$ "
have "hh holomorphic_on UNIV"
  <proof>
then show "hh analytic_on UNIV"
  by (simp add: analytic_on_open)

```

Here are a few details of the proof that was skipped above. For a fixed  $n$  and complex number  $z$ , we show uniform convergence within a circle of radius 1. First, we need a bound on  $\prod_{i < n} |z' - w_i|$  for  $|z - z'| < 1$ :

```

have norm_p_bound: "norm (p n z')  $\leq q \ n * (1 + \text{norm } z) ^ n$ "
  if "dist z z'  $\leq 1$ " for n z z'
  <proof>

```

To apply the Weierstrass  $M$ -test to  $hh$ , exhibit a summable series  $M$  whose terms bound the finite approximants of the summation, which are given by  $h$ :

```

have "uniform_limit (cball z 1) ( $\lambda n. h \ n \ \text{coeff}$ ) hh sequentially"
  unfolding hh_def h_def
proof (rule Weierstrass_m_test)
  let ?M = " $\lambda n. (1 + \text{norm } z) ^ n / \text{fact } n$ "
  show "summable ?M"
  <proof>
  fix n z'
  assume "z'  $\in \text{cball } z \ 1$ "
  show "norm (coeff n * p n z')  $\leq ?M \ n$ "
  <proof>
qed

```

The hard part is over. Checking that the image of  $\{\zeta_\alpha : \alpha < \gamma\}$  under  $hh$  lies within  $D$  is easy, because for every complex number of the form  $w_n$ , the summation is actually finite.

```

have hh_eq_dd: "hh (w n) = dd n coeff" for n
  <proof>
then have "hh (w n)  $\in D$ " for n
  using DD_def dd_in_DD by fastforce

```

Last comes the diagonalisation argument, showing that the function just constructed is indeed new:

```

have "hh (w n)  $\neq f$  ( $\eta \ n$ ) (w n)" for n
  using DD_def dd_in_DD g_def hh_eq_dd by auto
then show " $\forall \beta \in \text{elts } \gamma. hh \neq f \ \beta$ "
  by (metis  $\eta \ \text{bij\_betw\_imp\_surj\_on imageE}$ )

```

#### 4.4 Concluding the Proof

About 20 lines of boilerplate are needed to code up the formal application of transfinite recursion to the construction formalised above. Skipping over this,

let's look at the conclusion of the proof. We finally have a family of analytic functions  $\{f_\beta : \beta < \omega_1\}$  satisfying (2) above.

We now show that for all  $z$ , the set  $\{f_\beta(z) : \beta < \omega_1\}$  is countable by writing  $z = \zeta_\alpha$  for some  $\alpha < \gamma$ . The point is that

$$\{f_\beta(\zeta_\alpha) : \beta < \omega_1\} = \{f_\beta(\zeta_\alpha) : \beta \leq \alpha\} \cup \{f_\beta(\zeta_\alpha) : \alpha < \beta\}$$

and that both parts of the union are countable: the first is a collection below  $\alpha$ , a countable ordinal, and the second is a subset of  $D$ , a countable set. In the proof below,  $?B$  is the set  $\{\beta : \alpha < \beta < \gamma\}$ , while  $elts (succ \alpha)$  is the set of all ordinals up to  $\alpha$ .

**show** *?thesis*

**proof**

**let**  $?F = "f \text{ ' } elts \ \omega 1"$

**have** *"countable (( $\lambda f. f z$ ) '  $f$  '  $elts \ \omega 1$ )"* **for**  $z$

**proof -**

**obtain**  $\alpha$  **where**  $\alpha: " \zeta \ \alpha = z " " \alpha \in elts \ \omega 1 " " Ord \ \alpha "$

**by** (*meson Ord\_ $\omega 1$  Ord\_in\_Ord UNIV\_I  $\zeta$  bij\_betw\_iff bijections*)

**let**  $?B = "elts \ \omega 1 - elts (succ \ \alpha)"$

**have** *eq: "elts  $\omega 1 = elts (succ \ \alpha) \cup ?B"$*

**using**  $\alpha$  **by** (*metis Diff\_partition Ord\_ $\omega 1$  OrdmemD less\_eq\_V\_def succ\_le\_iff*)

**have** *"( $\lambda f. f z$ ) '  $f$  '  $?B \subseteq D"$*

**using**  $\alpha$  *inD* **by** *clarsimp (meson Ord\_ $\omega 1$  Ord\_in\_Ord Ord\_linear)*

**then** **have** *"countable (( $\lambda f. f z$ ) '  $f$  '  $?B)$* "

**by** (*meson countable D countable\_subset*)

**moreover** **have** *"countable (( $\lambda f. f z$ ) '  $f$  '  $elts (succ \ \alpha)$ )"*

**by** (*simp add:  $\alpha$  less\_ $\omega 1$ \_imp\_countable*)

**ultimately** **show** *?thesis*

**using** *eq* **by** (*metis countable\_Un\_iff image\_Un*)

**qed**

**then** **show** *"Wetzel ?F"*

**unfolding** *Wetzel\_def* **by** (*blast intro: anf*)

**show** *"uncountable ?F"*

**using** *Ord\_ $\omega 1$  countable\_iff\_less\_ $\omega 1$  countable\_image\_inj\_eq injf* **by** *blast*

**qed**

**qed**

The following corollary to the two cases summarises the equivalence between the Wetzel property and the negation of CH:

**theorem** *Erdos\_Wetzel: " $C\_continuum = \aleph 1 \longleftrightarrow (\exists F. Wetzel \ F \wedge uncountable \ F)$ "*

**by** (*metis C\_continuum\_ge Erdos\_Wetzel\_CH Erdos\_Wetzel\_nonCH less\_V\_def*)

## 5 Discussion

The formalisation of mathematics within proof assistants is being pursued actively. There are too many contributions to list, but notable recent ones include the Buzzard–Commelin–Massot formalisation of perfectoid spaces [3] and

the striking progress on the Liquid Tensor Experiment, which is led by Johan Commelin.<sup>5</sup> Both of these involve formalising the sophisticated work of Fields Medallist Peter Scholze, using Lean.

The small example reported here cannot be compared with such accomplishments, but shines a light on an issue that makes mathematics difficult: its interconnectedness. The connections between number theory and complex analysis are well known. If your problem concerns itself with sets of permutations, you might suddenly find yourself needing group theory. If you are trying to count the elements of the group, you might find yourself requiring sophisticated combinatorial arguments. If you find yourself talking about countable sets (and countability is an everyday notion), you could find yourself wandering into the world of cardinals, then into the world of ordinals and then perhaps into the undecidable. To quote Wetzel himself, “once again a natural analysis question has grown horns” [6, p. 244].

I created the ZFC-in-HOL library in October 2019 with no specific application in mind but merely thinking that it might be useful occasionally to talk about really big sets. Then I decided to apply it to a project that had been proposed by Mirna Džamonja and Angeliki Koutsoukou-Argyraiki: to formalise some ordinal partition theory. This concerns advanced generalisations of Ramsey’s theorem; the formalisation [18] was complete by August 2020 and a paper is now available [4]. However, this work essentially belongs to straight set theory, borrowing little from Isabelle/HOL apart from natural numbers and lists.

The Wetzel example showed that there were a few gaps limiting the integration between that theory and Isabelle/HOL, above all, a more general definition of cardinality. It’s easy to define the cardinality of an arbitrary Isabelle/HOL set (i.e. having a type of the form *T set* as the least ordinal whose elements can be put into bijection with that set. The equivalent notion of cardinality for ZF sets is a trivial consequence.

The original AFP entry [19], published online in February 2022, represents about three weeks’ work. It begins with a variety of extensions to the Isabelle/HOL libraries, showing precisely how the original ZFC library needed to be extended. It includes an early and rather cumbersome definition of cardinality for HOL sets. In updated versions of the development,<sup>6</sup> the library material has been moved elsewhere. The latest version is 360 lines, 2833 tokens, which is not bad compared with the original text by Erdős [5], which is one and a half pages long. The exposition by Aigner and Ziegler [2] is two full pages long (67 lines, 1026 tokens). A crude comparison based on compressing the texts by gzip suggests a de Bruijn factor of under 2.3. For the original version, including the material later moved to libraries, the de Bruijn factor is 4.2.

---

<sup>5</sup> <https://tinyurl.com/5n8rh297>

<sup>6</sup> At [https://devel.isa-afp.org/entries/Wetzels\\_Problem.html](https://devel.isa-afp.org/entries/Wetzels_Problem.html). Future readers should be able to locate the original Wetzel entry via “Older releases” on the download page.

## 6 Conclusions

The intriguing solution to Wetzel’s problem by Erdős was easily formalised in Isabelle/HOL, using a library designed to integrate set theory with higher-order logic. The ease with which we can intermix analytic functions with ordinals gives reason to hope that harder mixed-domain problems will be amenable to formalisation without particular difficulties. It would be interesting to see this example tackled using other proof assistants.

**Acknowledgements** This work was supported by the ERC Advanced Grant ALEXANDRIA (Project GA 742178). Dmitriy Traytel provided a particularly slick formal axiomatisation of type  $V$ . Manuel Eberl provided a crucial tip for the CH case. Angeliki Koutsoukou-Argraki and the referees made insightful suggestions.

For the purpose of open access, the author has applied a Creative Commons Attribution (CC BY) licence to any Author Accepted Manuscript version arising from this submission.

## References

1. Agerholm, S.: A comparison of HOL-ST and Isabelle/ZF. In: Paulson, L.C. (ed.) Proceedings of the First Isabelle Users Workshop. pp. 53–70. Technical Report 379, Computer Laboratory, University of Cambridge (Sep 1995)
2. Aigner, M., Ziegler, G.M.: Proofs from THE BOOK. Springer, 6th edn. (2018)
3. Buzzard, K., Commelin, J., Massot, P.: Formalising perfectoid spaces. In: Certified Programs and Proofs. pp. 299–312. Association for Computing Machinery (2020), <https://doi.org/10.1145/3372885.3373830>
4. Džamonja, M., Koutsoukou-Argraki, A., Paulson, L.C.: Formalising ordinal partition relations using Isabelle/HOL. Experimental Mathematics (in press), online at <https://doi.org/10.1080/10586458.2021.1980464>
5. Erdős, P.: An interpolation problem associated with the continuum hypothesis. Michigan Mathematical Journal **11**(1), 9–10 (Mar 1964)
6. Garcia, S.R., Shoemaker, A.L.: Wetzel’s problem, Paul Erdős, and the continuum hypothesis: A mathematical mystery. Notices of the AMS **62**(3), 245–247 (Mar 2015)
7. Gordon, M.: Set theory, higher order logic or both? In: von Wright, J., Grundy, J., Harrison, J. (eds.) Theorem Proving in Higher Order Logics: TPHOLs ’96. pp. 191–201. LNCS 1125, Springer (1996)
8. Gunther, E., Pagano, M., Sánchez Terraf, P.: Formalization of forcing in Isabelle/zf. In: Peltier, N., Sofronie-Stokkermans, V. (eds.) Automated Reasoning — 10th International Joint Conference, IJCAR 2020. pp. 221–235. Springer (2020)
9. Gunther, E., Pagano, M., Terraf, P.S., Steinberg, M.: The independence of the continuum hypothesis in Isabelle/ZF. Archive of Formal Proofs (Mar 2022), [https://isa-afp.org/entries/Independence\\_CH.html](https://isa-afp.org/entries/Independence_CH.html), Formal proof development
10. Han, J.M., van Doorn, F.: A formal proof of the independence of the continuum hypothesis. In: 9th ACM SIGPLAN Conference on Certified Programs and Proofs. pp. 353–366. CPP 2020, Association for Computing Machinery (2020). <https://doi.org/10.1145/3372885.3373826>

11. Kunen, K.: Set Theory: An Introduction to Independence Proofs. North-Holland (1980)
12. Nipkow, T., Paulson, L.C., Wenzel, M.: Isabelle/HOL: A Proof Assistant for Higher-Order Logic. Springer (2002), online at <http://isabelle.in.tum.de/dist/Isabelle/doc/tutorial.pdf>
13. Obua, S.: Partizan games in Isabelle/HOLZF. In: Barkaoui, K., Cavalcanti, A., Cerone, A. (eds.) Theoretical Aspects of Computing - ICTAC 2006. pp. 272–286. Springer (2006)
14. Paulson, L.C.: The foundation of a generic theorem prover. *Journal of Automated Reasoning* **5**(3), 363–397 (1989). <https://doi.org/10.1007/BF00248324>, <https://arxiv.org/abs/cs/9301105>
15. Paulson, L.C.: The reflection theorem: A study in meta-theoretic reasoning. In: Voronkov, A. (ed.) Automated Deduction — CADE-18 International Conference. pp. 377–391. LNAI 2392, Springer (2002)
16. Paulson, L.C.: The relative consistency of the axiom of choice—mechanized using Isabelle/ZF. *LMS Journal of Computation and Mathematics* **6**, 198–248 (2003). <https://doi.org/10.1112/S1461157000000449>
17. Paulson, L.C.: Zermelo Fraenkel set theory in higher-order logic. *Archive of Formal Proofs* (Oct 2019), [http://isa-afp.org/entries/ZFC\\_in\\_HOL.html](http://isa-afp.org/entries/ZFC_in_HOL.html), formal proof development
18. Paulson, L.C.: Ordinal partitions. *Archive of Formal Proofs* (Aug 2020), [http://isa-afp.org/entries/Ordinal\\_Partitions.html](http://isa-afp.org/entries/Ordinal_Partitions.html), Formal proof development
19. Paulson, L.C.: Wetzel’s problem and the continuum hypothesis. *Archive of Formal Proofs* (Feb 2022), [https://isa-afp.org/entries/Wetzels\\_Problem.html](https://isa-afp.org/entries/Wetzels_Problem.html), Formal proof development
20. Paulson, L.C., Grabczewski, K.: Mechanizing set theory: Cardinal arithmetic and the axiom of choice. *Journal of Automated Reasoning* **17**(3), 291–323 (Dec 1996). <https://doi.org/10.1007/BF00283132>
21. Schleicher, D., Stoll, M.: An introduction to Conway’s games and numbers (2004). <https://doi.org/10.48550/ARXIV.MATH/0410026>, <https://arxiv.org/abs/math/0410026>
22. Wenzel, M.: Type classes and overloading in higher-order logic. In: Gunter, E.L., Felty, A. (eds.) Theorem Proving in Higher Order Logics: TPHOLs ’97. pp. 307–322. LNCS 1275, Springer (1997)
23. Werner, B.: Sets in types, types in sets. In: Abadi, M., Ito, T. (eds.) Theoretical Aspects of Computer Software. pp. 530–546. Springer (1997)